**pmaxima**

```
Maxima 5.27.0 http://maxima.sourceforge.net
using Lisp GNU Common Lisp (GCL) GCL 2.6.7 (a.k.a. GCL)
Distributed under the GNU Public License. See the file COPYING.
Dedicated to the memory of William Schelter.
```

# Droite d'Euler et cercle circonscrit

Le fichier `geo2d.mac` (voir à la fin du document) chargé ci-dessous contient quelques macros permettant de faire des calculs en géométrie euclidienne plane. Le source est en fin de document.

```
▷ load("geo2d.mac")$
```

L'utilisation, ici, se rapporte à la résolution d'un petit exercice autour de la droite d'Euler et du cercle circonscrit.

> Dans le plan $\mathscr{P}$, muni d'un repère orthonormé $(\Omega, \vec{\imath}, \vec{\jmath})$, construire les points $A(-3,1)$, $B(1,5)$, $C(3,-3)$.
>
> **1/** Écrire une équation de chacune des trois hauteurs du triangle $ABC$. Justifier que ces trois hauteurs sont concourantes en un point $H$ dont on précisera les coordonnées.
>
> **2/** Soit $\Delta_1$, $\Delta_2$, $\Delta_3$ les médiatrices respectives de $[BC]$, $[CA]$, $[AB]$. Écrire une équation de chacune de ces médiatrices. Retrouver par le calcul que ces trois médiatrices sont concourantes en un point $O$ dont on précisera les coordonnées.
>
> **3/** Déterminer les coordonnées de l'isobarycentre $G$ de $ABC$.
>
> **4/** Démontrer la relation :
> $$\overrightarrow{OH} = 3\overrightarrow{OG}$$
> En déduire l'alignement des points $O$, $G$, $H$.
>
> **5/** Écrire une équation du cercle $\mathscr{C}$ circonscrit au triangle $ABC$.
>
> **6/** Soit $H_3$, $H_2$ et $H_1$ les symétriques de l'orthocentre $H$ par rapport aux droites respectives $(AB)$, $(CA)$, $(BC)$. Soit $K_3$, $K_2$, $K_1$ les symétriques de l'orthocentre $H$ par rapport aux milieux respectifs de $[AB]$, $[CA]$, $[BC]$.
> Déterminer les coordonnées des six points $H_1$, $H_2$, $H_3$, $K_1$, $K_2$, $K_3$. Vérifier que ces six points appartiennent au cercle circonscrit au triangle $ABC$.

Nous définissons les points, il s'agit de listes de 2 réels.

```
▷ [A:[-3,1],B:[1,5],C:[3,-3]];
```

2 : $[[-3,1],[1,5],[3,-3]]$

Nous déterminons les équations des trois hauteurs du triangle.

```
▷ h:[hauteur(A,B,C),hauteur(B,C,A),hauteur(C,A,B)];
```

3 : $[x-4\,y+7=0,-3\,x+2\,y-7=0,x+y=0]$

Nous définissons $H$ comme étant l'intersection des deux premières hauteurs et nous vérifions qu'il appartient à la troisième.

```
▷ H:interdroite(h[1],h[2]);
```

4 : $\left[-\frac{7}{5}, \frac{7}{5}\right]$

```
▷ H sur h[3];
```

5 : **true**

Nous déterminons une équation des trois médiatrices du triangle $ABC$.

```
▷ m:[mediatrice(A,B),mediatrice(B,C),mediatrice(C,A)];
```

6 : $[x+y-2=0, x-4y+2=0, -3x+2y+2=0]$

Nous définissons $O$ comme étant l'intersection des deux premières médiatrices et nous vérifions qu'il appartient à la troisième.

```
▷ O:interdroite(m[1],m[2]);
```

7 : $\left[\frac{6}{5}, \frac{4}{5}\right]$

```
▷ O sur m[3];
```

8 : **true**

Voici le centre de gravité du triangle $ABC$, on l'obtient à l'aide d'une banale *combinaison linéaire* de listes.

```
▷ G:1/3*(A+B+C);
```

9 : $\left[\frac{1}{3}, 1\right]$

Après le calcul des vecteurs $\overrightarrow{OH}$ et $\overrightarrow{OG}$, vient le calcul du vecteur $\overrightarrow{OH} - 3\overrightarrow{OG}$ pour vérifier qu'il est bien nul.

```
▷ (OH:H-O,OG:G-O,OH-3*OG);
```

10 : $[0, 0]$

Nous déterminons une équation cartésienne du cercle de centre $O$ et passant par $A$, c'est à dire le cercle circonscrit au triangle $ABC$.

```
▷ c:eqc([O,A]);
```

11 : $5x^2 - 12x + 5y^2 - 8y - 78 = 0$

Voici les symétriques de $H$ par rapport aux côtés du triangle et la vérification qu'ils appartiennent au cercle.

```
▷ sd:[symd(H,[A,B]),symd(H,[C,A]),symd(H,[B,C])];
```

12 : $\left[\left[-\frac{13}{5}, \frac{13}{5}\right], \left[-\frac{179}{65}, -\frac{41}{65}\right], \left[\frac{409}{85}, \frac{251}{85}\right]\right]$

```
▷ map(lambda([t],t sur c),sd);
```

13 : [**true**, **true**, **true**]

Même chose avec les symétriques de $H$ par rapport aux milieux des côtés du triangle.

```
▷ sp:[symp(H,milieu(A,B)),symp(H,milieu(C,A)),symp(H,milieu(B,C))];
```

14 : $\left[\left[-\frac{3}{5}, \frac{23}{5}\right], \left[\frac{7}{5}, -\frac{17}{5}\right], \left[\frac{27}{5}, \frac{3}{5}\right]\right]$

```
▷ map(lambda([t],t sur c),sp);
```

15 : [**true**, **true**, **true**]

```
/* Vecteur directement orthogonal.                                          */
vdo(v) := [-v[2],v[1]];

/* Norme d'un vecteur.                                                      */
norme2(v) := v.v;
norme(v) := sqrt(v.v);

/* Détermination de deux points d'une droite à partir d'une équation.       */
deq(e) := block([a:coeff(lhs(e),x),b:coeff(lhs(e),y),p],
    p:subst(solve([e,b*x-a*y=0]),[x,y]),
    [p,p+[b,-a]]);

/* Détermination d'une équation d'une droite à partir de deux points.       */
eqd(d) := block([v:d[2]-d[1]],
    rat(part(content(vdo(v).(d[1]-[x,y]),x,y),2),y,x)=0 );

/* Milieu d'un segment.                                                     */
milieu(a,b) := 1/2*(a+b);


/* Projection orthogonale d'un point sur une droite définie par deux        */
/* points.                                                                  */
pied(m,d) := block([v:d[2]-d[1]],
    subst(solve([eqd([m,m+vdo(v)]),eqd(d)]),[x,y]));


/* Projection orthogonale d'un point sur une droite définie par une         */
/* équation cartésienne.                                                    */
projection(m,e) := pied(m,deq(e));

/* Hauteur du triangle abc issue de a.                                      */
hauteur(a,b,c) := eqd([a,pied(a,[b,c])]);

/* Médiatrice du segment [ab].                                              */
mediatrice(a,b) := eqd([1/2*(a+b),1/2*(a+b)+vdo(b-a)]);

/* Intersection de deux droites.                                            */
interdroite(a,b) := subst(solve([a,b]),[x,y]);

/* Équation d'un cercle connaissant le centre et un point.                  */
eqc(c) := rat(
    part(content(norme2(c[1]-[x,y])-norme2(c[1]-c[2]),x,y),2),y,x)=0;

/* Symétrique d'un point par rapport à un autre point.                      */
symp(m,p) := 2*p-m;

/* Symétrique orthogonal d'un point par rapport à une droite définie        */
```

```
/* par deux points.                                                      */
symd(m,d) := 2*pied(m,d)-m;

/* Opérateur infixé « sur » pour détecter l’appartenance d’un point à   */
/* un ensemble dont on connaît une équation.                             */
infix("sur");
"sur"(p,e) := block([f],
    f:subst([x=p[1],y=p[2]],lhs(e)),
    if f = 0 then true else false);
```